

# ARTINIAN GRADED ALGEBRAS, THE WEAK LEFSCHETZ PROPERTY AND THE POSTULATION HILBERT SCHEME

GUNTRAM HAINKE AND ALMAR KAID

## 1. INTRODUCTION

This is a summary of the third tutorial handed out at the CoCoA summer school 2005. Our aim here is to investigate the weak Lefschetz property in the study of Artinian algebras. Moreover we provide the necessary CoCoA (version 4.6) code which can be used for further experiments. We thank Holger Brenner, Martin Kreuzer and Juan Migliore for helpful discussions.

Let  $K$  be an infinite field and  $P := K[x_0, \dots, x_n]$  the standard graded polynomial ring over  $K$ . Furthermore, let  $I \subseteq P$  be a homogeneous ideal and denote by  $A := P/I$  the corresponding quotient  $P$ -algebra. We recall that  $A$  is *Artinian* if every descending chain of ideals in  $A$  is eventually stationary.

To start with, we give some well-known reformulations of the definition of an Artinian algebra (compare also [5, Proposition 3.7.1] and [6, Proposition 5.6.30]).

**Proposition 1.1.** *Let  $I \subseteq P = K[x_0, \dots, x_n]$  be a homogeneous ideal and  $A := P/I$ . The following conditions are equivalent:*

- (1)  $A$  is Artinian.
- (2)  $\dim_K A < \infty$ .
- (3)  $\dim A = 0$  (*Krull dimension*), i.e. every prime ideal in  $A$  is maximal.
- (4) For any term ordering  $\sigma$  of  $P$  and for all  $i \in \{0, \dots, n\}$  there is an  $n_i \in \mathbb{N}$  such that  $x_i^{n_i} \in \text{LT}_\sigma(I)$ .

*Proof.* (1)  $\Rightarrow$  (3). Let  $\mathfrak{p} \subseteq A$  be a prime ideal. We show that the integral domain  $B := A/\mathfrak{p}$  is a field. So let  $x \in B$ ,  $x \neq 0$ . By the descending chain condition there exists some  $k \in \mathbb{N}$  such that  $Bx^k = Bx^{k+1}$ . But that means  $x^k = yx^{k+1}$  for some  $y \in B$  and hence  $1 = yx$ .

(3)  $\Rightarrow$  (2). Assume that every prime ideal in  $A$  is maximal. Since  $A$  is Noetherian, it has only finitely many minimal prime ideals (cf. [6, Proposition 5.6.15(b)]), i.e. the topological space  $X := \text{Spec } A = \{\mathfrak{m}_1, \dots, \mathfrak{m}_r\} = \{P_1, \dots, P_r\}$  is a discrete point set. In particular, we have  $\Gamma(\{P_i\}, \mathcal{O}_X) = A_{\mathfrak{m}_i}$  for  $i = 1, \dots, r$  and therefore  $A = \Gamma(X, \mathcal{O}_X) \cong A_{\mathfrak{m}_1} \times \dots \times A_{\mathfrak{m}_r}$ . Hence we can assume that  $A$  is even a local ring with maximal ideal  $\mathfrak{m}$ . The quotient algebra  $A/\mathfrak{m}$  is a finite dimensional vector space by the field theoretic version of Hilbert's Nullstellensatz (cf. [5, Theorem 2.6.6(b)]). Therefore  $\mathfrak{m}^j/\mathfrak{m}^{j+1} \cong \mathfrak{m}^j \otimes_A (A/\mathfrak{m})$  is finite dimensional too for  $j \geq 1$ . Since  $\mathfrak{m}^k = (0)$  for a  $k \in \mathbb{N}$ , it follows from the exact sequence

$$0 \longrightarrow \mathfrak{m}^{j+1} \longrightarrow \mathfrak{m}^j \longrightarrow \mathfrak{m}^j/\mathfrak{m}^{j+1} \longrightarrow 0$$

by descending induction that all  $\mathfrak{m}^j$  are finite dimensional vector spaces.

(2)  $\Rightarrow$  (1). By choosing an appropriate  $K$ -basis for each ideal in a chain  $q_0 \supset q_1 \supset \dots$  it follows that there are at most  $\dim_K A$  proper inclusions.

(2)  $\Rightarrow$  (4). Since  $A = P/I$  is finite dimensional, there exists for each  $i \in \{0, \dots, n\}$  some  $n_i \in \mathbb{N}$  such that  $x_i^{n_i} + \sum_{j=0}^{n_i-1} a_j x_i^j = 0$  holds in  $A$  for certain  $a_j \in K$ . In  $P$ , this translates to  $x_i^{n_i} + \sum_{j=0}^{n_i-1} a_j x_i^j \in I$  and therefore  $x_i^{n_i} \in I$  because  $I$  is homogeneous. In particular,  $x_i^{n_i} \in \text{LT}_\sigma(I)$ .

(4)  $\Rightarrow$  (2). By Macaulay's Basis Theorem (cf. [5, Theorem 1.5.7]),  $\mathbb{T}^n \setminus \text{LT}_\sigma(I)$  constitutes a  $K$ -basis for  $A$ , where  $\mathbb{T}^n$  denotes the monoid of terms of  $P$  (i.e. all monomials). As a subset of the finite set  $\{x_0^{b_0} \dots x_n^{b_n} \mid b_i \leq n_i\}$ , it is finite.  $\square$

Note that the proposition remains valid if the ideal  $I$  is not homogeneous. Indeed this assumption was used only in the implication "(2)  $\Rightarrow$  (4)"; in the general case one passes to the leading term ideal. Condition (3) of Proposition 1.1 allows a quick check with CoCoA whether  $A = P/I$  is Artinian:

```
Define IsArtinian(I)
Return Dim(CurrentRing()/I)=0;
EndDefine;
```

Since by Proposition 1.1 the graded  $P$ -algebra  $A$  is a finite dimensional vector space, it is of the form

$$A = K \oplus A_1 \oplus \dots \oplus A_s$$

for some  $s \geq 0$ . The *socle* of  $A$ , denoted  $\text{soc } A$ , is the annihilator of  $\mathfrak{m} := (x_1, \dots, x_n)$ , i.e.

$$\text{soc } A := ((0) :_A \mathfrak{m}) = \{x \in A : \mathfrak{m} \cdot x = 0\}.$$

The number  $s = \max\{i \mid A_i \neq 0\}$  is called the *socle degree* of  $A$ . Obviously,  $A_s \subseteq \text{soc } A$ . We recall that the *h-vector* of  $A$  is the vector  $(1, \dim_K A_1, \dots, \dim_K A_s)$ . The following one-liner computes the socle degree of  $A$ :

```
Define SocDeg(I)
Return Len(HVector(CurrentRing()/I))-1;
EndDefine;
```

**Definition 1.2.** Let  $I \subseteq P$  be a homogeneous ideal such that  $A = P/I$  is Artinian.

- (1)  $A$  is called *Gorenstein* if its socle is a 1-dimensional  $K$ -vector space.
- (2) The algebra  $A$  is called *level* if its socle is precisely  $A_s$ , where  $s$  is the socle degree of  $A$ .

This definition coincides with other definitions of Gorenstein found in the literature. Clearly Gorenstein algebras are level algebras. With the help of the function `SocDeg(I)` defined above we are able to define two CoCoA functions `IsArtinianGorenstein(I)` and `IsLevel(I)` which check whether  $A$  has the properties in question and return the corresponding Boolean value.

```
Define IsArtinianGorenstein(I)
If IsArtinian(I)=False Then
Return False;
EndIf;
Return Last(HVector(CurrentRing()/I))=1;
EndDefine;
```

```
Define IsLevel(I)
If IsArtinian(I)=False Then
Return False;
EndIf;
Socle:=I:Ideal(Indets());
DimSocle:=Len(QuotientBasis(I))
-Len(QuotientBasis(Socle));
Return DimSocle=Last(HVector(CurrentRing()/I));
EndDefine;
```

**Definition 1.3.** Let  $I \subseteq P$  be a homogeneous ideal such that  $A = P/I$  is Artinian.

- (1)  $A$  has the *weak Lefschetz property (WLP)* if, for a *general* linear form  $\ell \in P_1$ , the multiplication maps  $\mu_\ell : A_i \rightarrow A_{i+1}$  have maximal rank for all  $i \in \{1, \dots, s-1\}$ , i.e. the maps  $\mu_\ell$  are either injective or surjective.
- (2)  $A$  has the *strong Lefschetz property (SLP)* if, for every  $d \geq 1$  and *every general* form  $f \in P_d$ , the multiplication maps  $\mu_f : A_i \rightarrow A_{i+d}$  have maximal rank for all  $i \in \{1, \dots, s-1\}$ .

**Remark 1.4.** We have to explain the term *general* in Definition 1.3. The space of linear forms  $P_1$  is a  $(n+1)$ -dimensional vector space over  $K$  and can be identified with the  $K$ -valued points of an affine space  $\mathbb{A}^{n+1}$  endowed with the Zariski topology. To say that a property depending on a linear form holds *generally* or *generically* means that there exists a *non-empty* Zariski-open subset  $U \subseteq \mathbb{A}^{n+1}$  such that for  $\ell \in U$  the property holds. The best approximation for a general linear form which CoCoA provides is given by the command `Randomized(DensePoly(1));`.

**Remark 1.5.** We want to point out that our definition of SLP is often called the *maximal rank property* (cf. [9, Definition 1.1]). The usual notion is: the Artinian algebra  $A$  has SLP if for every general linear form  $\ell$  and every  $d \geq 1$  the maps  $\mu_{\ell^d}$  have maximal rank for all  $i \in \{1, \dots, s-1\}$ . One observes that this property implies the maximal rank property by semicontinuity.

The following proposition shows that the h-vector of an Artinian  $P$ -algebra having the weak Lefschetz property has a certain shape.

**Proposition 1.6.** *Let  $A = P/I$  be an Artinian algebra with socle degree  $s$  which has WLP. Then the Hilbert function  $\text{HF}_A$  satisfies*

$$1 = \text{HF}_A(0) < \text{HF}_A(1) < \dots < \text{HF}_A(t) \geq \text{HF}_A(t+1) \geq \dots \geq \text{HF}_A(s)$$

for some  $t \geq 1$ .

*Proof.* Let  $\ell$  be a generic linear form. Assume that  $A_t \xrightarrow{\cdot\ell} A_{t+1}$  is surjective for a  $t < s$ . It is enough to show that  $A_{t+1} \xrightarrow{\cdot\ell} A_{t+2}$  is surjective as well. Since  $A$  is a standard graded algebra, we can write every  $x \in A_{t+2}$  as  $x = \sum_{i=1}^r a_i b_i$  with  $a_i \in A_1$  and  $b_i \in A_{t+1}$ . By assumption each  $b_i$  can be written as  $b_i = \ell c_i$  with  $c_i \in A_t$ . Hence  $x = \ell \sum_{i=1}^r a_i c_i$  and  $a_i c_i \in A_{t+1}$  for  $i = 1, \dots, r$ .  $\square$

To decide with the help of CoCoA whether  $A$  has one of these properties, we first write a function `HasMaxRank(I,D)`. This function checks whether the map  $\mu_f$  has maximal rank for a general (or rather randomized) form  $f \in P_d$ .

```

Define HasMaxRank(I,D)
  L:=Randomized(DensePoly(D));
  J:=Ideal([L])+I;
  HJ:=Hilbert(CurrentRing()/J);
  HI:=Hilbert(CurrentRing()/I);
  K:=0;
  While (EvalHilbertFn(HI,K)>0) Do
    Dif:=Max(EvalHilbertFn(HI,K+D)
      -EvalHilbertFn(HI,K),0);
    If EvalHilbertFn(HJ,K+D)<>Dif Then
      Return False;
    EndIf;
    K:=K+1;
  EndWhile;
Return True;
EndDefine;

```

This function uses the exactness of the sequence

$$(P/I)_k \xrightarrow{\cdot F} (P/I)_{k+d} \longrightarrow (P/(I+(F)))_{k+d} \longrightarrow 0.$$

Now we can check whether  $A$  has WLP or SLP:

```

Define HasWLP(I)
Return HasMaxRank(I,1);
EndDefine;

Define HasSLP(I)
  Result:=True;
  For D:=1 To SocDeg(I) Do
    Result:=Result AND HasMaxRank(I,D);
  EndFor;
Return Result;
EndDefine;

```

Finally, we will define the function `GenLinForm(N)` which approximates a general linear form  $\ell \in P_1$  with integer coefficients between  $-N$  and  $N$ :

```

Define GenLinForm(N);

```

```
Return Sum([Rand(-N,N)*L | L In Indets()]);
EndDefine;
```

## 2. WEAK LEFSCHETZ PROPERTY AND THE POSTULATION HILBERT SCHEME

In this section we follow essentially [8, Paragraph after Remark 3.3 and Example 3.4]. Let  $Z$  denote a zero-dimensional subscheme of  $\mathbb{P}^n$  with coordinate ring  $R_Z := P/I_Z$ . Then the Hilbert function of  $Z$  is defined as the Hilbert function of  $R_Z$ . If  $\lambda \in P_1$  is a general linear form, the Artinian graded algebra  $A_Z := P/(I_Z + (\lambda))$  is called the *Artinian reduction* of  $R_Z$ . In particular, the Betti diagrams of  $R_Z$  and  $A_Z$  coincide and we have

$$\mathrm{HF}_{A_Z}(i) = \mathrm{HF}_{R_Z}(i) - \mathrm{HF}_{R_Z}(i-1) = \Delta \mathrm{HF}_{R_Z}(i),$$

i.e. the Hilbert function (or h-vector) of  $A_Z$  is given by the first difference of the Hilbert function of  $R_Z$  (cf. [7, Proposition 4.3]). *We say that  $Z$  has the weak Lefschetz property if its Artinian reduction  $A_Z$  has this property.*

Now, we fix a Hilbert function  $H$  of a zero-dimensional scheme and the corresponding h-vector  $h = (a_0, \dots, a_s)$  respectively. Next we consider the “postulation” Hilbert scheme  $\mathrm{Hilb}^H(\mathbb{P}^n)$  parameterizing all zero-dimensional subschemes of  $\mathbb{P}^n$  having the given Hilbert function  $H$  (cf. [4, Definition 5.20]). Since one can compute the degree  $d$  of such a scheme by  $d = \sum_{i=0}^s a_i$ ,  $\mathrm{Hilb}^H(\mathbb{P}^n)$  is sitting inside the “punctual” Hilbert scheme  $\mathrm{Hilb}^d(\mathbb{P}^n)$  which parameterizes all zero-dimensional subschemes of  $\mathbb{P}^n$  of degree  $d$ . For our further investigations in this section it is not necessary to pass over to the closure of  $\mathrm{Hilb}^H(\mathbb{P}^n)$  inside  $\mathrm{Hilb}^d(\mathbb{P}^n)$  as it was done in [8].

In the sequel we will study the behavior of the weak Lefschetz property on some irreducible components of  $\mathrm{Hilb}^H(\mathbb{P}^n)$ .

**Proposition 2.1.** *The weak Lefschetz property is an open property on  $\mathrm{Hilb}^H(\mathbb{P}^n)$ , i.e. in any irreducible component of  $\mathrm{Hilb}^H(\mathbb{P}^n)$  there exists a Zariski-open subset (possibly empty) which corresponds to zero-dimensional schemes having WLP.*

*Proof.* For every point  $x \in \mathrm{Hilb}^H(\mathbb{P}^n)$  we consider the maps

$$(A_{x,\lambda})_i := (P/(I_x + (\lambda)))_i \xrightarrow{\cdot \ell} (A_{x,\lambda})_{i+1} := (P/(I_x + (\lambda)))_{i+1}$$

for  $\lambda, \ell \in P_1$  and for a fixed value  $i \in \mathbb{Z}$ , where  $I_x$  denotes the homogeneous ideal parameterized by  $x$ . The parameter  $x, \lambda, \ell$  vary in

$Hilb^H(\mathbb{P}^n) \times P_1 \times P_1$ . The vector space  $(A_{x,\lambda})_i$  corresponds to a point in the Grassmanian  $\mathbb{G}_i := Grass(\dim_K P_i - a_i, P_i)$ . More precisely, by the universal property of Grassmanians, there exists a morphism  $\psi_i : Hilb^H(\mathbb{P}^n) \times P_1 \rightarrow \mathbb{G}_i$  such that  $V_{\psi_i(x,\lambda)} = I_x + (\lambda)$ . For  $y \in \mathbb{G}_i$  there exists a basis  $v_1, \dots, v_{\dim_K P_i}$  of  $P_i$  and an open neighborhood  $y \in U \subseteq \mathbb{G}_i$  such that the images of  $v_1, \dots, v_{a_i}$  form a basis of  $P_i/V_z$  for all  $z \in U$  (compare [12, Beispiel 1.B.8]). This induces for every  $z \in U$  a  $K$ -vector space isomorphism  $\theta_{i,z} : K^{a_i} \rightarrow P_i/V_z$  by sending  $e_k \mapsto \overline{v_k}$ . We consider now for  $(x, \lambda, \ell) \in Hilb^H(\mathbb{P}^n) \times P_1 \times P_1$  the commutative diagram

$$\begin{array}{ccc} K^{a_i} & \xrightarrow{\cdot \tilde{\ell}} & K^{a_{i+1}} \\ \theta_{i,\psi_i(x,\lambda)} \downarrow & & \downarrow \theta_{i+1,\psi_{i+1}(x,\lambda)} \\ (A_{x,\lambda})_i & \xrightarrow{\cdot \ell} & (A_{x,\lambda})_{i+1}. \end{array}$$

Here the linear map

$$\tilde{\ell} = \theta_{i+1,\psi_{i+1}(x,\lambda)}^{-1} \circ \ell \circ \theta_{i,\psi_i(x,\lambda)}$$

depends algebraically on  $(x, \lambda, \ell) \in Hilb^H(\mathbb{P}^n) \times P_1 \times P_1$ . As the maximal rank property of  $\tilde{\ell}$  (hence of  $\ell$ ) can be checked by looking at minors, there exists an Zariski open subset  $W_i \subseteq Hilb^H(\mathbb{P}^n) \times P_1 \times P_1$  such that  $\tilde{\ell} = \tilde{\ell}(i, x, \lambda, \ell)$  has the maximal rank property if and only if  $(x, \lambda, \ell) \in W_i$ . Hence there exists also an open subset  $W \subseteq Hilb^H(\mathbb{P}^n) \times P_1 \times P_1$  such that  $\tilde{\ell} = \tilde{\ell}(x, \lambda, \ell)$  has the maximal rank property for all  $i$  if and only if  $(x, \lambda, \ell) \in W$ . Since the projection  $p_1 : Hilb^H(\mathbb{P}^n) \times P_1 \times P_1 \rightarrow Hilb^H(\mathbb{P}^n)$  is open, we get an open subset  $p_1(W)$  in  $Hilb^H(\mathbb{P}^n)$ . For an element  $x \in Hilb^H(\mathbb{P}^n)$  the Artinian reduction of  $P/I_x$  has WLP if and only if there exists  $\lambda \in P_1$  and  $\ell \in P_1$ , such that  $(P/(I_x + (\lambda)))_i \xrightarrow{\cdot \ell} (P/(I_x + (\lambda)))_{i+1}$  has maximal rank for all  $i$ . This is exactly the case if  $x \in p_1(W)$ .  $\square$

In the following two examples we will work in the Hilbert scheme  $Hilb^H(\mathbb{P}^3)$  where  $H$  is the Hilbert function which corresponds to the h-vector  $(1, 3, 6, 9, 11, 11, 11)$ .

**Example 2.2.** First we consider the ideal

$$I_1 := (x_1^3, x_1^2 x_2^2, x_1^2 x_2 x_3^2, x_3^5) + (x_1, x_2, x_3)^7 \subseteq R := K[x_1, x_2, x_3].$$

We realize the situation in CoCoA via the commands

```

Use R:=Q[x[1..3]];
M:=Ideal(Indets());
I_1:=Ideal(x[1]^3,x[1]^2x[2]^2,x[1]^2x[2]x[3]^2,x[3]^5)
+M^7;

```

Further, by using

```
IsArtinian(R/I_1);
```

we see that the corresponding quotient algebra  $A_1 := R/I_1$  is Artinian. The command

```
HVector(R/I_1);
```

yields that the h-vector of  $A_1$  equals  $h = (1, 3, 6, 9, 11, 11, 11)$ . Next we compute the Betti diagram of  $A_1$  via

```
Bettidiagram(R/I_1);
```

and get

	0	1	2	3
0:	1	-	-	-
1:	-	-	-	-
2:	-	1	-	-
3:	-	1	1	-
4:	-	2	2	1
5:	-	-	-	-
6:	-	10	22	11
Tot:	1	14	25	12

From this Betti diagram we see that  $A_1$  has a socle element in degree 4 (compare for instance [2, Exercise 3.3.26] and use  $\text{soc } A_1 \cong \text{Hom}_P(K, A_1)$ ). Since a socle element will be annihilated by any linear form  $\ell$ , the map

$$(A_1)_4 \xrightarrow{\cdot \ell} (A_1)_5$$

has a non-zero kernel, i.e. is not injective. But both of these vector spaces are of dimension 11, hence the map above is not surjective either. Therefore, the Artinian algebra  $A_1$  has not WLP.

Since  $I_1$  is a monomial ideal, we lift  $I_1$  to the vanishing ideal  $I_{Z_1}$  of a (reduced) zero-dimensional subscheme  $Z_1$  in  $\mathbb{P}^3$  (compare for example [10] for an explicit description of the lifting procedure). Moreover,  $A_1$  equals the Artinian reduction of the homogeneous coordinate ring of

$Z_1$ , i.e.  $Z_1$  corresponds to a point in the Hilbert scheme  $\text{Hilb}^H(\mathbb{P}^3)$ , where  $H = 1, 4, 10, 19, 30, 41, 52, 52, \dots$

Since the graded Betti numbers are upper semicontinuous (cf. for instance [11, Lemma 1.2]) with respect to the flat family  $\pi : X \rightarrow \text{Hilb}^H(\mathbb{P}^3)$  (where  $X$  is the universal family and the fiber  $X_y = \pi^{-1}(y)$ ,  $y \in \text{Hilb}^H(\mathbb{P}^3)$ , is the zero-dimensional scheme parameterized by  $y$  having Hilbert function  $H$ ), there exists an Zariski-open subset  $U$  of the irreducible component containing  $Z_1$  where the graded Betti numbers are minimal, i.e. the graded Betti numbers of the general element in the irreducible component of  $Z_1$  can only go down. Via

$\text{Res}(R/I_1)$ ;

we compute the minimal graded free resolution of  $A_1$

$$0 \rightarrow R(-7) \oplus R(-11) \rightarrow R(-5) \oplus R(-6) \oplus R(-8) \rightarrow R(-3) \oplus R(-4) \oplus R(-5) \oplus R(-10) \rightarrow R$$

explicitly. The copy of  $R(-7)$  in the last free module of this resolution indicates the socle element in degree 4. That this copy can not vanish for the generic element in the Hilbert scheme component of  $Z_1$  follows from the following lemma.

**Lemma 2.3.** *Let  $x$  be a point in  $\text{Hilb}^H(\mathbb{P}^3)$  represented by the zero-dimensional scheme  $Z_x$  and let*

$$\mathbb{F}_\bullet : 0 \rightarrow \mathbb{F}_3 \rightarrow \mathbb{F}_2 \rightarrow \mathbb{F}_1 \rightarrow \mathbb{F}_0 = P \rightarrow P/I_x \rightarrow 0$$

with  $\mathbb{F}_i = \bigoplus_j P(-j)^{\beta_{i,j}}$  be the minimal graded free resolution of the coordinate ring  $P/I_x$  of  $Z_x$ . Then the minimal graded free resolution of the general element in the irreducible component of  $x$  is obtained from  $\mathbb{F}_\bullet$  by the cancellation of some ghost-terms (i.e. a ghost-term is a pair of copies  $P(-d)$ ,  $d \in \mathbb{Z}$ , in two consecutive free modules  $\mathbb{F}_i$  and  $\mathbb{F}_{i-1}$ ,  $i = 2, 3$ ).

*Proof.* By the upper semicontinuity of the graded Betti numbers the graded Betti numbers of the general element can only go down from  $\mathbb{F}_\bullet$ . We can assume that the resolution has no ghost-terms, since cancellation of these terms does not change the situation numerically. We have to show that the minimal free resolution of  $P/I_x$  is also the minimal free resolution of the general element in the irreducible component of  $x$ . Further, we consider instead of the free modules  $\mathbb{F}_i$  the locally free sheaves  $\tilde{\mathbb{F}}_i = \bigoplus_j \mathcal{O}_{\mathbb{P}^3}(-j)^{\beta_{i,j}}$  for  $i = 0, \dots, 3$  which decompose as a direct sum of line bundles on  $\mathbb{P}^3$ . In particular, we have  $\text{HF}_{A_{Z_x}}(d) = \sum_{i=0}^3 h^0(\tilde{\mathbb{F}}_i(d))$  ( $h^0$  denotes the dimension of the vector

space of global sections). Now we assume that terms in the minimal resolution of the general element vanish and denote these for each  $i$  by  $\mathcal{F}_i = \bigoplus_k \mathcal{O}_{\mathbb{P}^3}(-k)^{n_i} \subseteq \tilde{\mathbb{F}}_i$  for  $i = 1, 2, 3$ . Since the Hilbert function does not change, we have  $h^0(\mathcal{F}_1(n)) + h^0(\mathcal{F}_3(n)) = h^0(\mathcal{F}_2(n))$  for all  $n \in \mathbb{Z}$ . Let  $m$  denote the minimum of all twists which occur in  $\mathcal{F}_1, \mathcal{F}_2, \mathcal{F}_3$ . Let this minimum be in  $\mathcal{F}_3$ . If we tensor  $\mathcal{F}_1, \mathcal{F}_2, \mathcal{F}_3$  with  $\mathcal{O}_{\mathbb{P}^3}(m)$  we get  $h^0(\mathcal{F}_1(m)) \geq 0$ ,  $h^0(\mathcal{F}_2(m)) = 0$  and  $h^0(\mathcal{F}_3(m)) \neq 0$ . But this gives a contradiction. Also in the other cases we get a contradiction.  $\square$

By the Betti diagram of  $Z_1$  above and Lemma 2.3 we see that for every  $x \in U$  the Artinian reduction of the corresponding zero-dimensional scheme  $Z_x$  has also a socle element in degree 4 and therefore  $Z_x$  fails to have WLP. Therefore by Proposition 2.1 no element in the irreducible component containing  $Z_1$  has WLP since two non-empty open subsets of the irreducible component have a non-empty intersection. In particular, the open set referred to in Proposition 2.1 is empty.

**Example 2.4.** For our second example, we construct a set  $Z_2$  of 52 points lying on a curve  $C$  of degree 11 in  $\mathbb{P}^3$  which realizes the truncated Hilbert function, i.e. satisfies  $\text{HF}_{Z_2}(t) = \min\{52, \text{HF}_C(t)\}$ . This is achieved by constructing points which are “general enough” and is done in two steps:

*Step 1.* Take  $D$  to be the union of a line  $L$  and 8 general points  $P_1, \dots, P_8$  in  $\mathbb{P}^3$ . We realize the corresponding vanishing ideal of  $D$  via

```
Use P:=Q[x[0..3]];
Points:=NewList(8);
For I:=1 To 8 Do
  Points[I]:=[Rand(-10,10) | J In 1..4];
EndFor;
IPoints:=IdealOfProjectivePoints(Points);
IL:=Ideal(GenLinForm(10),GenLinForm(10));
ID:=Intersection(IPoints,IL);
```

Now we can link  $D$  via a complete intersection curve  $X$  of type  $(3, 4)$  to a smooth curve  $C$  which is by the degree formula  $\deg(X) - \deg(D) = \deg(C)$  (cf. [7, §8]) of degree 11. To do this, the following function will be very useful:

```
Define GenRegSeq(I,L)
  M:=Ideal(Indets());
  Seq:=[];
```

```

N:=Ideal(0);
Foreach T In L Do
  J:=Intersection(M^T,I);
  D:=Deg(Head(MinGens(J)));
  If D<>T Then
    PrintLn"No regular sequence of this type possible";
    Return;
  EndIf;
  L2:=[F In MinGens(J) | Deg(F)=D];
  S:=Sum([Rand(-1000,1000)*F | F In L2]);
  If N:Ideal(S)<>N Then
    PrintLn"No regular sequence of this type possible";
    Return;
  EndIf;
  Append(Seq,S);
  N:=Ideal(Seq);
EndForeach;
Return Seq;
EndDefine;

```

The function `GenRegSeq(I,L)` takes an ideal  $I$  and a list of degrees  $L$  and computes a regular sequence of homogeneous polynomials in  $I$  whose degrees are given by  $L$ . If this is not possible it conveys this information. Using this function we compute the vanishing ideal of  $C$  with

```

IX:=Ideal(GenRegSeq(ID, [3,4]));
IC:=IX:ID;

```

If we compute the minimal graded free resolution of the homogeneous coordinate ring  $P/I_C$  of the curve  $C$  by

```

Res(P/IC);

```

we get

```

0 --> P^2(-6) --> P(-3)(+)P(-4)(+)P(-5) --> P
-----

```

i.e.  $C$  is arithmetically Cohen-Macaulay, since we have  $\text{pd}(P/I_C) = 2 = \text{codim}(I_C)$  where  $\text{codim } I = \min\{\text{ht}(\mathfrak{p}) : I \subseteq \mathfrak{p} \text{ minimal}\}$ .

We check with CoCoA that  $C$  is a smooth curve by computing the singular locus:

```

Define IsSmooth(I)
  J:=Jacobian(Gens(I));

```

```

L:=List(Minors(1,J));
SingLoc:=Radical(Ideal(L)+I);
Return SingLoc=Ideal(Indets());
EndDefine;

```

```

IsSmooth(IC);

```

*Step 2.* Now that we have the ideal of 8 generic points on  $C$ , we add suitable hyperplane sections of  $C$ . This is done by taking four hyperplane sections which contain two of the eight points (giving  $4 \cdot 9 = 36$  new points on  $C$ ) and one hyperplane section containing three of the eight points (giving a total of  $8+36+8=52$  points on  $C$ ). Here is our CoCoA code:

```

IPoints12:=IdealOfProjectivePoints([Points[1],Points[2]]);
GensIPoints12:=Gens(IPoints12);
L:=NewList(4);
For J:=1 To 4 Do
  L[J]:=Rand(-100,100)*GensIPoints12[1]
  +Rand(-100,100)*GensIPoints12[2];
EndFor;

```

```

Q:=NewList(4);
For J:=1 To 4 Do
  Q[J]:=Colon(IC+Ideal(L[J]),IPoints);
EndFor;

```

```

IPoints123:=IdealOfProjectivePoints([Points[1],Points[2],
Points[3]]);
HIPoints123:=Comp(Gens(IPoints123),1);
QQ:=Colon(IC+Ideal(HIPoints123),IPoints123);

```

Hence we get the homogeneous vanishing ideal of our (reduced) zero-dimensional scheme via

```

IZ_2:=IntersectionList([IPoints,Q[1],Q[2],Q[3],Q[4],QQ]);

```

We check by

```

Hilbert(P/IZ_2);

```

that the Hilbert function of the coordinate ring of  $Z_2$  equals:

```

H(0) = 1
H(1) = 4
H(2) = 10

```

$H(3) = 19$   
 $H(4) = 30$   
 $H(5) = 41$   
 $H(t) = 52$  for  $t \geq 6$

-----  
 If we compute the Artinian reduction  $A_2$  of  $Z_2$  via

`IA_2:=IZ_2+Ideal(GenLinForm(10));`

we can check by using our function `HasWLP(I)` that  $A_2$  and therefore the scheme  $Z_2$  has the weak Lefschetz property. It follows from Proposition 2.1 that the general element in the irreducible component containing  $Z_2$  has WLP. Indeed, if we compute the Betti diagram

	0	1	2	3
0:	1	-	-	-
1:	-	-	-	-
2:	-	1	-	-
3:	-	1	-	-
4:	-	1	2	-
5:	-	-	-	-
6:	-	11	22	11
Tot:	1	14	24	11

-----  
 of  $P/I_{Z_2}$  with

`BettiDiagram(P/IZ_2);`

we see that it can not be a specialization of the Betti Diagram corresponding to  $Z_1$  given in Example 2.2 and vice versa, i.e. the schemes  $Z_1$  and  $Z_2$  belong to different irreducible components of  $Hilb^H(\mathbb{P}^3)$ .

### 3. HILBERT FUNCTIONS OF COMPLETE INTERSECTIONS AND WLP

The starting point of our investigations in this section is the following theorem.

**Theorem 3.1.** *Every Artinian complete intersection in  $K[x_0, x_1, x_2]$  has the weak Lefschetz property.*

*Proof.* See [3, Theorem 2.3] or see [1, Corollary 2.4] for a more conceptual proof. □

The Artinian reduction of a complete intersection in  $\mathbb{P}^3$  has by Theorem 3.1 the weak Lefschetz property. We will show in the sequel that there still may exist a zero-dimensional scheme in  $\mathbb{P}^3$  with a Hilbert function of a complete intersection, but its Artinian reduction does not have WLP.

**Example 3.2.** First, we consider the complete intersection of type  $(2, 2, 4)$  given by  $I := (x_0^2, x_1^2, x_2^4) \subset P := K[x_0, x_1, x_2, x_3]$  and denote the corresponding zero-dimensional scheme by  $Z_1 = V_+(I)$ . Its Artinian reduction

$$A_{Z_1} \cong K[x_0, x_1, x_2]/(x_0^2, x_1^2, x_2^4)$$

has by Theorem 3.1 the weak Lefschetz property. We can also verify this by using CoCoA and our function `HasWLP(I)`. We compute the h-vector of  $A_{Z_1} := R/I$  via

```
Use R:=Q[x[0..2]];
I:=Ideal(x[0]^2,x[1]^2,x[2]^4);
HVector(R/I);
```

and get  $h = (1, 3, 4, 4, 3, 1)$ .

Next, we construct a zero-dimensional scheme in  $\mathbb{P}^3$  having the same h-vector as the complete intersection above. We start with a generic complete intersection in  $\mathbb{P}^2$  of type  $(4, 4)$  containing the points  $P_1 = (1 : 0 : 0)$  and  $P_2 = (0 : 1 : 0)$  (one gets this points with CoCoA by `GenericPoints(2)`). Hence, we use the commands

```
Use R:=Q[x[0..2]];
IP_1P_2:=IdealOfProjectivePoints(GenericPoints(2));
ICi:=Ideal(GenRegSeq(IP_1P_2,[4,4]));
```

and compute the vanishing ideal  $I_{\mathbb{X}}$  of the pointset  $\mathbb{X}$  which consists of the residual 14 points via

```
IX:=ICi:IP_1P_2;
```

If we compute the Hilbert function of  $R/I_{\mathbb{X}}$  with

```
Hilbert(R/IX);
```

we get

```
H(0) = 1
H(1) = 3
H(2) = 6
H(3) = 10
H(4) = 13
H(t) = 14 for t >= 5
```

-----

i.e.  $\Delta \text{HF}_{\mathbb{X}} = (1, 2, 3, 4, 3, 1)$ . Next, we embed  $\mathbb{X}$  in  $\mathbb{P}^3$  via the canonical inclusion  $\mathbb{P}^2 \rightarrow \mathbb{P}^3$  and consider the zero dimensional scheme  $Z_2 := \mathbb{X} \cup \{P, Q\}$ , where  $P$  and  $Q$  are general points in  $\mathbb{P}^3$ . Our CoCoA code for this is:

```
GensIX:=Gens(IX);
Use P:=Q[x[0..3]];
IXinP3:=Ideal(BringIn(GensIX))+ Ideal(x[3]);
IP:=Ideal(GenLinForm(20),GenLinForm(20),GenLinForm(20));
IQ:=Ideal(GenLinForm(20),GenLinForm(20),GenLinForm(20));
IZ_2:=Intersection(IXinP3,IP,IQ);
```

Note that in this code we can not just use the implemented function `GenericPoints(2)` to obtain  $P$  and  $Q$  since this returns by default the points  $(1 : 0 : 0 : 0)$  and  $(0 : 1 : 0 : 0)$  which are not generic anymore in this context. If we compute further the Artinian reduction  $A_{Z_2}$  of  $Z_2$  and its h-vector by

```
IA_Z_2:=IZ_2 + Ideal(GenLinForm(20));
HVector(P/IA_Z_2);
```

we notice that the Artinian algebras  $A_{Z_1}$  and  $A_{Z_2}$  have the same h-vector. Now we obtain the Betti diagram of  $Z_2$  via

```
BettiDiagram(P/IZ_2);
```

and get

	0	1	2	3
0:	1	-	-	-
1:	-	2	1	-
2:	-	1	2	1
3:	-	2	2	-
4:	-	1	2	1
5:	-	-	1	1
Tot:	1	6	8	3

-----

From this diagram we see that  $A_{Z_2}$  has a socle element in degree 2, i.e. the linear map

$$(A_{Z_2})_2 \xrightarrow{\cdot \ell} (A_{Z_2})_3$$

is not injective for any linear form  $\ell$ . Since  $\dim_K((A_{Z_2})_2) = 4 = \dim_K((A_{Z_2})_3)$  this map fails to be surjective too, i.e.  $Z_2$  does not have the weak Lefschetz property.

Here, we can not decide whether these two zero-dimensional schemes belong to different irreducible components of  $Hilb^H(\mathbb{P}^3)$  where  $H$  is the Hilbert function corresponding to the h-vector  $(1, 3, 4, 4, 3, 1)$ . Because if we look at the Betti Diagram

	0	1	2	3
0:	1	-	-	-
1:	-	2	-	-
2:	-	-	1	-
3:	-	1	-	-
4:	-	-	2	-
5:	-	-	-	1
Tot:	1	3	3	1

of  $Z_1$ , we can not exclude whether  $Z_1$  is a specialization of  $Z_2$ .

#### REFERENCES

- [1] H. Brenner and A. Kaid. Syzygy bundles on  $\mathbb{P}^2$  and the weak Lefschetz property. *ArXiv*, 2006.
- [2] W. Bruns and J. Herzog. *Cohen-Macaulay Rings, Revised edition*. Cambridge University Press, 1998.
- [3] T. Harima, J. C. Migliore, U. Nagel, and J. Watanabe. The weak and strong Lefschetz properties for Artinian  $k$ -algebras. *J. Algebra*, 262:99–126, 2003.
- [4] A. Iarrobino and V. Kanev. *Power Sums, Gorenstein Algebras, and Determinantal Loci*. Springer, Berlin Heidelberg, 1999.
- [5] M. Kreuzer and L. Robbiano. *Computational Commutative Algebra*, volume 1. Springer, Berlin, 2000.
- [6] M. Kreuzer and L. Robbiano. *Computational Commutative Algebra*, volume 2. Springer, Berlin, 2005.
- [7] J. Migliore. Experiments in commutative algebra and algebraic geometry - fourth international CoCoA school on computer algebra, Porto Conte, May 23-26. In this volume, 2005.
- [8] J. Migliore. The geometry of Hilbert functions. *Arxiv*, 2005.
- [9] J. Migliore and R. M. Miró-Roig. Ideals of general forms and the ubiquity of the weak Lefschetz property. *J. Pure App. Alg.*, 182(1):79–107, 2003.
- [10] J. Migliore and U. Nagel. Lifting monomial ideals. *Comm. Algebra*, 28:5679–5701, 2000.

- [11] A. Ragusa and G. Zappalá. On the reducibility of the postulation Hilbert scheme. *Rendiconti del Circolo Matematico di Palermo*, Serie II, Tomo LIII:401–406, 2004.
- [12] U. Storch and H. Wiebe. *Lehrbuch der Mathematik, Band 4*. Spektrum Akademischer Verlag, 2001.

FAKULTÄT FÜR MATHEMATIK, UNIVERSITÄT BIELEFELD, D-33501 BIELEFELD,  
GERMANY

*E-mail address:* [ghainke@math.uni-bielefeld.de](mailto:ghainke@math.uni-bielefeld.de)

DEPARTMENT OF PURE MATHEMATICS, UNIVERSITY OF SHEFFIELD, HOUNSFIELD  
ROAD, SHEFFIELD S3 7RH, UNITED KINGDOM

*E-mail address:* [A.Kaid@sheffield.ac.uk](mailto:A.Kaid@sheffield.ac.uk)